
Цель лабораторной работы: изучение концепций и освоение технологии структурного программирования, приобретение навыков структурного программирования на языке Турбо Паскаль рекуррентных вычислений.

Задание на программирование: используя технологию структурного программирования разработать программу решения индивидуальной задачи, содержащей 3 вида циклических управляющих структур: Цикл - Пока (с предусловием), Цикл - До (с постусловием), Цикл - Для (с параметром).

Порядок выполнения работы:

1) Получить у преподавателя индивидуальное задание. Выполнить постановку задачи: сформулировать условие, определить входные и выходные данные.
2) Разработать математическую модель.
3) Построить схему алгоритма, последовательно используя для решения задачи все три циклические управляющие структуры (операторы while, repeat…until, for).
4) Составить программу на языке Турбо Паскаль.
5) Входные данные вводить с клавиатуры по запросу.
6) Выходные данные выводить на экран в развернутой форме с пояснениями.
7) Проверить и продемонстрировать преподавателю работу программы на полном наборе тестов, в том числе с ошибочными входными данными.
8) Оформить отчет о лабораторной работе в составе: постановка задачи, математическая модель, схема алгоритма решения, текст программы, контрольные примеры.



Найти предел произведения  для последовательности {Yn}, пользуясь рекуррентной формулой
Y1 = 1; Yn = n*(Yn-1 + 1); n = 2, 3, 4,…
Вычисления прекратить при выполнении условия |Yn  - Yn-1| <ε.

Примеры программ

Пример 1
Формулировка задания:
Пользуясь рекуррентной формулой Yi = Yi-1 + Yi-3 * Yi-3, где i=3,4,…n, для заданного значения n вычислить Yn, если известны Y0, Y1, Y2.

Program Recur_1;
{Пользуясь рекуррентной формулой y(i)=y(i-1)+y(i-3)^2
 где i=3,4,...,n для заданного значения n вычислить y(n),
 если известны y(0), y(1), y(2).
 Входные данные:  y(0),
                  y(1),
                  y(2) - значения начальных элементов
                     n - номер искомого значения.
 Выходное данное: y(n) - значение элемента с номером n
}

Var
  y0, y1, y2,                   {исходные данные}
  y:LongInt;                    {результат}
  a, b, c,                      {вспомогательные переменные}
  i,                            {номер текущего элемента}
  n:Integer;                    {номер искомого элемента}
  txt:string;                   {вид заголовка пользовательского окна}
Begin
  WriteLn('Пользуясь рекуррентной формулой y(i)=y(i-1)+y(i-3)^2');
  WriteLn('где i=3,4,...,n для заданного значения n вычислить y(n),');
  WriteLn('если известны y(0), y(1), y(2).');
  Write('Задайте значения y(0), y(1) и y(2) -> ');
  ReadLn(y0,y1,y2);
  Write('Задайте значение n -> ');
  ReadLn(n);
  a:=y0;
  b:=y1;
  c:=y2;
{Решение с использованием управляющей структуры While-Do}
  i:=3;
  While i<=n
  Do Begin
       y:=y2+y0*y0;  {значение очередного элемента последовательности}
       y0:=y1;
       y1:=y2;
       y2:=y;
       i:=i+1;
     End;
  WriteLn('While:  значение элемента последовательности y(',n,')=',y);
{Решение с использованием управляющей структуры Repeat-Until}
  y0:=a;
  y1:=b;
  y2:=c;
  i:=3;
  Repeat
    y:=y2+y0*y0;    {значение очередного элемента последовательности}
    y0:=y1;
    y1:=y2;
    y2:=y;
    i:=i+1;
  Until i>n;
  WriteLn('Repeat: значение элемента последовательности y(',n,')=',y);
{Решение с использованием управляющей структуры For-To-Do}
  y0:=a;
  y1:=b;
  y2:=c;
  For i:=3 To n
  Do Begin
       y:=y2+y0*y0;  {значение очередного элемента последовательности}
       y0:=y1;
       y1:=y2;
       y2:=y;
     End;
  WriteLn('For:    значение элемента последовательности y(',n,')=',y);
  ReadLn;
End.
Пример 2
Формулировка задания:
Последовательность функций Yn = Yn(X), где 0 < X < 1 определяется следующим образом: Y1 = (X + 1) / 2; Yn = (X + Yn-1 + 1) / 2; n = 2,3,4…
Для заданного значения X найти предел последовательности, принимая за таковой значение Yn, удовлетворяющее условию |Yn – Yn-1| < , где 0 <  <= 0.1.

Program Recur_2;
{Последовательность функций yn= yn(x), где 0 < x < 1 определяется
 следующим образом: y(1)=(x+1)/2; y(n)=(x+y(n-1)+1)/2; n=2,3,4...
 Для заданного x найти предел последовательности, принимая за таковой
 значение y(n), удовлетворяющее условию |y(n) - y(n-1)|<eps,
 где 0 < eps <= 0.1.
}
Uses Crt;       {Подключение модуля}

Var
    x,          {входное данное - значение аргумента}
  eps,          {входное данное - точность вычисления}
   y1,          {текущее значение функции}
   yn:Real;     {значение предела последовательности}
    n:Integer;  {номер предельного элемента}
Begin
  ClrScr;
  WriteLn('Последовательность функций yn=yn(x), где 0<x<1 определяется');
  WriteLn('следующим образом: y(1)=(x+1)/2; y(n)=(x+y(n-1)+1)/2; n=2,3,4...');
  WriteLn('Для заданного x найти предел последовательности, принимая за таковой');
  WriteLn('значение y(n), удовлетворяющее условию |y(n)-y(n-1)|<eps');
  Write('где 0 < eps <= 0.1');
  Write(' Введите значение точности: eps=');
  ReadLn(eps);
  If (eps<=0)OR(eps>0.1)
  Then Begin
         Write(' Ошибка ввода! Значение eps должно быть <=0.1 и >0');
         ReadLn;
         Exit;
       End;
  Write(' Введите значение аргумента:  x=');
  ReadLn(x);
  If (x<=0)OR(x>=1)
  Then Begin
         Write(' Ошибка ввода! Значение x должно быть >0 и <1');
         ReadLn;
         Exit;
       End;
{Цикл While}
  n:=1;
  y1:=0;
  yn:=(x+1)/2;
  While (Abs(yn - y1)>eps)AND(n<MAXINT)
{наибольшее значение типа Integer=32767}
  Do Begin
       y1:=yn;
       yn:=(x+y1+1)/2;
       n:=n+1;
     End;
  If n=MAXINT
  Then WriteLn(' Точность не достигнута.')
  Else WriteLn(' Для цикла While  предел y(',n,')=',yn:1:7,', n=',n);
{Цикл Repeat}
  n:=1;
  y1:=0;
  yn:=(x+1)/2;
  Repeat
    y1:=yn;
    n:=n+1;
    yn:=(x+y1+1)/2;
  Until (Abs(yn-y1)<eps)OR(n>=MAXINT);
  If n=MAXINT
  Then WriteLn(' Точность не достигнута.')
  Else WriteLn(' Для цикла Repeat предел y(',n,')=',yn:1:7,', n=',n);

{Цикл For}
  yn:=(x+1)/2;
  for n:=2 To MAXINT
  Do Begin
       y1:=yn;
       yn:=(x+y1+1)/2;
       If(Abs(yn-y1)<eps)
       Then Break;
     End;
  If n=MAXINT
  Then WriteLn(' Точность не достигнута.')
  Else WriteLn(' Для цикла For    предел y(',n,')=',yn:1:7,', n=',n);
  ReadLn;
End.

